Полностью пустая форма, одна процедура при создании на сервере вот код

```1c
    &НаСервере
    Процедура ПриСозданииНаСервере(Отказ, СтандартнаяОбработка)

    ТипыРеквизита = Новый Массив;
    ТипыРеквизита.Добавить(Тип("ДинамическийСписок"));
    ОписаниеТиповДляРеквизита = Новый ОписаниеТипов(ТипыРеквизита);
    ДинамическийСписок = Новый РеквизитФормы("Список", ОписаниеТиповДляРеквизита,,"Список",ЛОЖЬ);
    ДобавляемыеРеквизиты = Новый Массив;
    ДобавляемыеРеквизиты.Добавить(ДинамическийСписок);
    ИзменитьРеквизиты(ДобавляемыеРеквизиты);

    РеквизитДинамическийСписок = ЭтаФорма["Список"];
    РеквизитДинамическийСписок.ПроизвольныйЗапрос = ИСТИНА;
    РеквизитДинамическийСписок.ТекстЗапроса = "ВЫБРАТЬ Ложь КАК Поле1,
    | ИСТИНА КАК Поле2,
    | ЛОЖЬ КАК Поле3,
    | ИСТИНА КАК Поле4,
    | ЛОЖЬ КАК Поле5";

    ТаблицаФормы = Элементы.Вставить("Список",Тип("ТаблицаФормы"));
    ТаблицаФормы.ПутьКДанным = "Список";

    Для А = 1 По 5 Цикл

    Заголовок = "Поле" + А;
    ИмяКолонки = "Поле" + А;

    НоваяКолонкаТаблицы = Элементы.Добавить("Список" + ИмяКолонки, Тип("ПолеФормы"), ТаблицаФормы);
    НоваяКолонкаТаблицы.Заголовок = Заголовок;
    НоваяКолонкаТаблицы.ПутьКДанным = "Список" + "." + ИмяКолонки;

    КонецЦикла;

    // Условное оформление
    Для А = 1 По 5 Цикл

    //ОфЛожь = ЭтотОбъект["Список"].УсловноеОформление.Элементы.Добавить();
    ОфЛожь = УсловноеОформление.Элементы.Добавить();
    Отбор = ОфЛожь.Отбор.Элементы.Добавить(Тип("ЭлементОтбораКомпоновкиДанных"));
    Отбор.ЛевоеЗначение = Новый ПолеКомпоновкиДанных("СписокПоле" + А);
    Отбор.ВидСравнения = ВидСравненияКомпоновкиДанных.Равно;
    Отбор.ПравоеЗначение = Ложь;
    ПолеОформления = ОфЛожь.Поля.Элементы.Добавить();
    ПолеОформления.Поле = Новый ПолеКомпоновкиДанных("СписокПоле" + А);
    ОфЛожь.Оформление.УстановитьЗначениеПараметра("Текст", "Неправда");

    //ОфИстина = ЭтотОбъект["Список"].УсловноеОформление.Элементы.Добавить();
    ОфИстина = УсловноеОформление.Элементы.Добавить();
    Отбор = ОфИстина.Отбор.Элементы.Добавить(Тип("ЭлементОтбораКомпоновкиДанных"));
    Отбор.ЛевоеЗначение = Новый ПолеКомпоновкиДанных("СписокПоле" + А);
    Отбор.ВидСравнения = ВидСравненияКомпоновкиДанных.Равно;
    Отбор.ПравоеЗначение = Истина;
    ПолеОформления = ОфИстина.Поля.Элементы.Добавить();
    ПолеОформления.Поле = Новый ПолеКомпоновкиДанных("СписокПоле" + А);
    ОфИстина.Оформление.УстановитьЗначениеПараметра("Текст", "Правда");

    КонецЦикла;

    КонецПроцедуры
```

Пробовал и условное оформление формы и самого списка.

```1c
    //ОфЛожь = ЭтотОбъект["Список"].УсловноеОформление.Элементы.Добавить();
    ОфЛожь = УсловноеОформление.Элементы.Добавить();
```

Не работает ни так ни так.

П.С. Версия платформы: 8.3.11.3034.
---
Разобрался.

Нужно добавлять не в

```1c
    ЭтотОбъект["Список"].УсловноеОформление.Элементы.Добавить();
```

а в

```1c
    ЭтотОбъект[ИмяСписка].КомпоновщикНастроек.ФиксированныеНастройки.УсловноеОформление.Элементы.Добавить();
```
---
Попробуй

```1c
    ОфЛожь = Список.УсловноеОформление.Элементы.Добавить();

    Отбор = ОфЛожь.Отбор.Элементы.Добавить(Тип("ЭлементОтбораКомпоновкиДанных"));
    Отбор.Использование = Истина;
    Отбор.ЛевоеЗначение = Новый ПолеКомпоновкиДанных("СписокПоле" + А);
    Отбор.ВидСравнения = ВидСравненияКомпоновкиДанных.Равно;
    Отбор.ПравоеЗначение = Ложь;

    Элемент = ОфЛожь.Оформление.Элементы[7];
    Элемент.Использование = Истина;
    Элемент.Значение = "Неправда";
```
---
Использование = Истина по отладчику по умолчанию стоит.
---
Может быть, просто я уже как-то привык)
И да. я ошибся при обращении к элементу оформления с параметром "Текст". Это не 7 индекс, а 21:

```1c
    Элемент = ОфЛожь.Оформление.Элементы[21];
    Элемент.Использование = Истина;
    Элемент.Значение = "Неправда";
```
---
Можешь добавить для определенного поля, что-то типа так:

```1c
    ОфПоле = ЭлементОформления.Поля.Элементы.Добавить();
    ОфПоле.Использование = Истина;
    ОфПоле.Поле = Новый ПолеКомпоновкиДанных("СписокПоле" + А);
```
---
Разобрался.

Нужно добавлять не в

```1c
    ЭтотОбъект["Список"].УсловноеОформление.Элементы.Добавить();
```

а в

```1c
    ЭтотОбъект[ИмяСписка].КомпоновщикНастроек.ФиксированныеНастройки.УсловноеОформление.Элементы.Добавить();
```